Herramientas para aplicar la estadística y el control de calidad en las empresas

Herramientas estadísticas (EHE)

Guillermo Sánchez (http://diarium.usal.es/guillermo)

Presentación el 2015-04-23 :

Los datos

Distribución de la población española por pueblos

In[1]:=
```
poblacionciudades = CityData[#, "Population"] & /@ CityData[{All, "Spain"}];
```

In[2]:=
```
Length[poblacionciudades]
```

Out[2]=
```
8066
```

In[3]:=
```
{Mean[#], StandardDeviation[#], Skewness[#], Kurtosis[#],
   Quantile[#, .6], InterquartileRange[#]} &[poblacionciudades] // N
```

Out[3]=
```
{5564.95 people , 47334.1 people , 47.9311, 3010.32, 980. people , 2168. people }
```

In[4]:=
```
ListLogPlot[poblacionciudades]
```

Out[4]=



Agrupaciones

In[5]:=
```
Manipulate[Histogram[Drop[Drop[poblacionciudades, a], -b]],
 {{a, 500}, 0, 1000, 10}, {{b, 500}, 0, 1000, 100}, SaveDefinitions → True]
```

Out[5]=



Drop::normal : Nonatomic expression expected at position 1 in Drop[poblacionciudades, 500]. ≫

Drop::drop : Cannot drop positions −500 through −1 in Drop[poblacionciudades, 500]. ≫

Histogram::ldata : Drop[Drop[poblacionciudades, 500], −500] is not a valid dataset or list of datasets. ≫

Drop::normal : Nonatomic expression expected at position 1 in Drop[poblacionciudades, 500]. ≫

Drop::drop : Cannot drop positions −500 through −1 in Drop[poblacionciudades, 500]. ≫

Histogram::ldata : Drop[Drop[poblacionciudades, 500], −500] is not a valid dataset or list of datasets. ≫

¿Que se deduce de la representación de los datos?

Los datos multivariantes ¿Cómo representarlos?

In[53]:=
```
multidata = RandomVariate[MultivariatePoissonDistribution[1, {2, 3, 4}], 125]
```

Out[53]=
```
{{5, 6, 3}, {6, 2, 5}, {4, 3, 5}, {2, 4, 8}, {3, 3, 3}, {2, 7, 4},
 {4, 6, 5}, {2, 1, 5}, {2, 3, 5}, {3, 2, 3}, {8, 11, 8}, {3, 9, 2}, {3, 2, 2},
 {2, 3, 6}, {2, 2, 7}, {5, 6, 9}, {2, 4, 1}, {3, 1, 5}, {7, 9, 8}, {3, 3, 5},
 {2, 4, 1}, {1, 3, 3}, {1, 3, 7}, {6, 5, 8}, {6, 6, 8}, {4, 2, 3}, {0, 0, 8},
 {4, 5, 4}, {5, 3, 4}, {3, 5, 8}, {2, 4, 4}, {5, 6, 6}, {1, 2, 2}, {5, 2, 2},
 {3, 6, 8}, {3, 1, 3}, {2, 2, 3}, {2, 5, 8}, {4, 6, 4}, {3, 3, 7}, {3, 2, 4},
 {4, 6, 4}, {1, 2, 4}, {4, 5, 8}, {2, 2, 0}, {3, 4, 6}, {2, 3, 5}, {2, 3, 4},
 {2, 6, 7}, {4, 3, 4}, {2, 4, 5}, {2, 6, 5}, {4, 4, 5}, {6, 6, 8}, {1, 4, 7},
 {4, 3, 4}, {4, 0, 5}, {6, 2, 10}, {4, 6, 5}, {3, 2, 4}, {4, 6, 9}, {3, 8, 5},
 {3, 3, 4}, {2, 2, 3}, {1, 2, 4}, {2, 3, 4}, {4, 2, 4}, {3, 3, 6}, {3, 2, 4},
 {3, 1, 6}, {4, 3, 4}, {5, 3, 3}, {1, 1, 2}, {3, 5, 4}, {1, 2, 2}, {3, 4, 4},
 {1, 5, 2}, {2, 10, 8}, {5, 5, 5}, {3, 3, 2}, {3, 2, 4}, {4, 2, 8}, {1, 3, 8},
 {3, 2, 1}, {4, 2, 3}, {5, 0, 10}, {2, 3, 9}, {5, 11, 9}, {3, 5, 9}, {3, 4, 5},
 {2, 3, 5}, {4, 5, 7}, {2, 3, 5}, {4, 6, 3}, {0, 4, 3}, {6, 3, 7}, {5, 1, 4},
 {3, 6, 4}, {3, 4, 1}, {3, 0, 2}, {2, 5, 6}, {1, 5, 7}, {0, 0, 3}, {3, 4, 6},
 {6, 4, 4}, {3, 5, 4}, {1, 5, 6}, {2, 3, 6}, {2, 4, 5}, {8, 8, 6}, {8, 8, 10},
 {4, 8, 10}, {2, 2, 5}, {2, 4, 5}, {4, 9, 10}, {3, 3, 6}, {3, 2, 6}, {1, 1, 6},
 {0, 5, 3}, {4, 3, 4}, {3, 4, 9}, {0, 2, 2}, {6, 8, 7}, {3, 9, 6}, {3, 2, 6}}
```

In[54]:=
```
Show[{ListPointPlot3D[multidata],
  Graphics3D[{Red, PointSize[Large], Point[Mean[multidata]]}]}]
```

Out[54]=



In[55]:=
```
Histogram3D[multidata[[All, 1 ;; 2]], 5]
```

Out[55]=



In[56]:=
```
Row[{SmoothHistogram3D[multidata[[All, 1 ;; 2]], ImageSize → Medium],
  SmoothDensityHistogram[multidata[[All, 1 ;; 2]], ImageSize → Medium]}]
```

Out[56]=





Aquí un conjunto de datos trivariantes son visualizados de dos formas, en un sistema de coordenadas x,y,z y un plano {x, y} y la coordenada z como burbujas en cuyo tamaño (la coordenada z es el radio de la burbuja).

In[57]:=
```
GraphicsRow[{ListPointPlot3D[multidata, PlotStyle → PointSize[.025]],
  BubbleChart[multidata, BubbleScale → "Diameter",
   BubbleSizes → {.025, .075}]}, ImageSize → Medium]
```

Out[57]=



Un conjunto de datos cuadridimensionales pueden ser representados en un gráfico de burbujas (la cuarta dimensión es el color).

In[10]:=
```
BubbleChart3D[RandomReal[10, {25, 4}], ColorFunction → "AvocadoColors"]
```

Out[10]=



Los datos son aleatorios pero frecuentemente siguen distribuciones de probabilidad

In[11]:=
```
Manipulate[
 Histogram[RandomVariate[NormalDistribution[0, 1], a], Automatic, "PDF", Epilog →
   First@Plot[PDF[NormalDistribution[0, 1], x], {x, -4, 4}, PlotStyle → Red]],
 {{a, 1000, "Número de datos"}, 50, 10000}]
```

Out[11]=



La representación gráfica nos dirá mucho acerca de la distribución de probabiliad

In[12]:=
```
data = RandomVariate[d = NormalDistribution[12, 3], 30];
```

Los representamos y los comparamos con la función de densidad de una población normal.

In[13]:=
```
Show[
 Histogram[data, 5, "ProbabilityDensity"],
 Plot[PDF[NormalDistribution[12, 3], x], {x, 0, 20}, PlotStyle → Thick]]
```

Out[13]=



Test de hipótesis.-

Permiten saber si es razonable asumir que los datos siguen una determinada función de probabilidad

El planteamiento consiste en formular una hipoteis nula H_0: $\mu = \mu_0$ frente a otra alternativa (por ej.:H_1: $\mu \neq \mu_0$)
Para probar H_0 se toma una muestra aleatoria de la población . Se calcula una estadística de prueba apropiada, y después se rechaza o no la hipótesis nula H_0.

- Pueden cometerse dos tipos de errores:

  Error de tipo **I**.- α = P{error de tipo I}=P{rechazar H_0/ H_0es verdadera}
- Error de tipo **II**.- β = P{error de tipo II}=P{no rechazar H_0/ H_0es falsa}

A veces se utiliza el concepto de *potencia de la prueba, P,* que se define como
P = 1 - β = P{ rechazar H_0/ H_0es falsa},
es decir es la probabilidad de rechazar correctamente H_0

Comentarios

- α es conocido como el riesgo de fabricante, o del vendedor, ya que denota la probabilidad de rechazar un lote bueno, o de dar por insatisfactorio el funcionamiento de un equipo o proceso que realmente opera correctamente.
  β es conocido como el riesgo de comprador ya que denota la probabilidad de aceptar un producto con calidad deficiente o que se permita seguir operando un equipo o proceso que realmente opera incorrectamente, y por consiguiente fabricando lotes malos.
- Un procedimiento frecuente, para aplicar una prueba de hipótesis, es fijar el valor de α y despues diseñar un procedimiento para obtener un valor pequeño de β . El riesgo de β en general disminuye cuando el tamaño de la muestra aumenta y se controla indirectamente. Un procedimiento de este tipo, en el que se fija el valor α, tiene el problema de que solo nos indica si la hipotesis se admite o no, pero no nos proporciona información si en una prueba concreta el riesgo con el que se ha admitido (o rechazado) al hipotesis ha sido alto o bajo. Para evitar este problema es mas frecuente utilizar el criterio del P-valor que definimos como el nivel de significación mas pequeño con el que rechazariamos la hipótesis nula. Este es el criterio que aplicaremos en lo que sigue.

Test para comprobar si los datos se pueden ajustar a una determinada distibución

Aplicamos varios tests para ver la mejor función de distribución a la que se ajustan los datos

In[14]:=
```
ℋ = DistributionFitTest[data, Automatic, "HypothesisTestData"];
```

In[15]:=
```
ℋ["TestDataTable", All]
```

Out[15]=

	Statistic	P- Value
Anderson- Darling	0.302802	0.602498
Baringhaus- Henze	0.178803	0.507008
Cramér- von Mises	0.0390273	0.696079
Jarque- Bera ALM	1.31575	0.413957
Kolmogorov- Smirnov	0.0803866	0.896597
Kuiper	0.116025	0.867037
Mardia Combined	1.31575	0.413957
Mardia Kurtosis	−0.842636	0.399432
Mardia Skewness	0.555058	0.456258
Pearson χ^2	2.	0.849145
Shapiro- Wilk	0.963348	0.376096
Watson U^2	0.0354344	0.752032

Valores de P (P-Values) > 0.05 (como es el caso) indican que es razonable asumir que los datos proceden de una distribución normal.

In[16]:=
```
ProbabilityPlot[data, d]
```

Out[16]=



Herramientas estadísticas (EHE)

En http://www3.enusa.es/webMathematica/Estadistica/estadistica.htm pueden descargarse ficheros en Word que contienen algunas normas ISO aplicables al control de calidad, con ejemplos en excel. Además se incluyen unas serie de cálculos que pueden ejecutarse directamente desde el navegador.

Tecnicas de estimación y pruebas relativas a las media y varianza (ISO 2854-1976)

El procedimiento habitual en control de procesos y el control de calidad ese inferir una serie de conclusiones respecto de la población a partir de una muestra de la misma. Con frecuencia supondremos que se usan *muestras aleatorias.* Es decir, se tomará una muestra $x_1,..., x_n$, de manera que las observaciones $\{x_i\}$ se distribuyan independientemente.

Habitualmente supondremos que el muestreo es con reposición, aunque no lo sea, pues el tratamiento estadístico es mas sencillo y cuando la muestra es muy pequeña comparada con la población no afectara practicamente a las conclusiones.

Pruebas ISO 2854 - 576

Las hipótesis mas usuales a comprobar para poblaciones y muestras son relativas a las estimaciones de las medias y varianzas.En las tablas se resumen los hipótesis más usuales a comprobar.Las letras mayusculas (A, a H) identifican el procedimiento aplicable de la ISO 2854 - 576

MEDIAS	Varianza conocida	Varianza desconocida
Comparación de la media con un valor dado	A	A'
Estimación de la media	B	B'
Comparación de dos medias	C	C'
Estimación de la diferencia de dos medias	D	D'

Varianzas	Prueba
Comparación de varianzas con un valor dado	E
Estimación de la varianza	F
Comparación de dos varianzas	G
Estimación de la razón de dos varianzas	H

Las hipotesis para la varianza son directamente extensibles a las desviaciones estandar sin mas que tener en cuenta que $\sigma = \sqrt{\sigma^2}$

En todos los test es necesario fijar previamente el nivel de significación α (usualmente se elige α = 0.01, o α = 0.05)

A.- Comparación de μ con un valor dado m_0 con σ^2 conocida.

Sea μ la media de una población (no necesariamente normal), de varianza σ^2 conocida. Tomemos una muestra con media x y tamaño n.Comparamos μ con un valor dado m_0 para los siguientes casos :

- Hipotesis nula H_0: $\mu = m_0$. La hipotesis se rechaza si:

$$|\overline{x} - m_0| > z_{1-\alpha/2} \frac{\sigma}{\sqrt{n}} \tag{1}$$

- Hipotesis nula H_0: $\mu \geq m_0$. La hipotesis se rechaza si:

$$\overline{x} < m_0 - z_{1-\alpha} \frac{\sigma}{\sqrt{n}} \tag{2}$$

- Hipotesis nula H_0: $\mu \leq m_0$. La hipotesis se rechaza si:

$$\overline{x} > m_0 + z_{1-\alpha} \frac{\sigma}{\sqrt{n}} \tag{3}$$

Ejemplo 1

Uno de los criterios de aceptacion establecidos para utilizar botellas de vidrio, usadas para embasar bebidas gaseososas, es que la resistencia media a la presión interna sea mayor de 175 psi. Una embotelladora recibe una muestra de 25 botellas obteniendo los siguientes valores, en psi : {199, 198, 201, 185, 182, 172, 170, 177, 189, 198, 173, 177, 171, 179, 168, 185, 180, 169, 175, 191, 181, 182, 182, 187, 179}. El fabricante informa que su proceso presenta una desviación estándar σ = 10 psi.¿Pueden reutilizarse las botellas de este fabricante?.

Comenzamos por elaborar una lista con los datos del ejemplo, que llamamos , de presión interna pI, y calculamos su media que llameremos $\overline{x}_{PI}$:

```
In[17]:= pI = {199, 198, 201, 185, 182, 172, 170, 177, 189, 198, 173, 177,
    171, 179, 168, 185, 180, 169, 175, 191, 181, 182, 182, 187, 179};
```

```
In[18]:= x̄pI = Mean[pI]
Out[18]= 182
```

Se trata de probar la hipótesis con $\mu \geq m_0$ con m_0 =175 psi. Para ellos aplicamos (3) fijando un nivel de significación α = 5 %.

Ahora sustituimos en (3), donde m_0= 175 psi y σ = 10 psi .

```
In[19]:= z[c_] := Quantile[NormalDistribution[0, 1], c]
```

```
In[20]:= n = Length[pI]
Out[20]= 25
```

$$\text{In[21]:= } LS = 175 + \frac{10}{\sqrt{n}} z[0.95]$$

```
Out[21]= 178.29
```

In[22]:= $\overline{X}_{pI} \geq LS$

Out[22]= True

Conclusión : La hipotesis nula H_0 : $\mu \leq m_0$ es rechazada, por tanto se puede considerar admisible la hipotesis alternativa H_1 : $\mu > m_0$. Es decir se pueden admitir las botellas para su reutilización

Nota : El test anterior está incorporado en ref/Ztest

In[23]:= `ZTest[pI, 10^2, 175, AlternativeHypothesis → "Less"]`

Out[23]= 0.999767

A'.- Comparación de μ con un valor dado m_0 con σ^2 desconocida.

Sea μ la media de una población (no necesariamente normal), de varianza conocida σ^2.Tomemos una muestra con media $\overline{x}$ y desviación estandar muestral S, tamaño *n*. Comparamos μ con un valor dado m_0 para los siguientes casos :

- Hipotesis nula H_0: $\mu = m_0$. La hipotesis se rechaza si:

$$\left| \overline{x} - m_0 \right| > t_{(1-\alpha/2,\, \nu)} \frac{S}{\sqrt{n}} \qquad (4)$$

- Hipotesis nula H_0: $\mu \geq m_0$. La hipotesis se rechaza si:

$$\overline{x} < m_0 - t_{(1-\alpha,\, \nu)} \frac{S}{\sqrt{n}} \qquad (5)$$

- Hipotesis nula H_0: $\mu \leqslant m_0$. La hipotesis se rechaza si:

$$\overline{x} > m_0 + t_{(1-\alpha,\, \nu)} \frac{S}{\sqrt{n}} \qquad (6)$$

Ejemplo 2

Repetir el ejercicio anterior asumiendo como estimador de varianza el de la muestra.

Empezaremos por calcular la desviación standardzde la lista pI (la media es la antes calculada)

In[24]:=
```
SpI = StandardDeviation[pI]
```

Out[24]= $3\sqrt{\frac{21}{2}}$

se trata de probar la hipótesis $\mu \geq m_0$ con m_0= 175 psi. Para elloaplicamos (5) fijando un nivel de significación α=5%.

Determinar el valor de $z_{1-\alpha}$

In[25]:=
```
t[c_] := Quantile[StudentTDistribution[Length[pI - 1]], c]
```

Ahora sustituimos en (5), donde m_0= 175 psi.

In[26]:= $\overline{x}_{pI} <= 175 + \frac{S_{pI}}{\sqrt{n}}\, t[0.95]$

Out[26]=
```
False
```

Conclusión : La hipotesis nula H_0 : $\mu \leq m_0$ es rechazada . Es admisible la hipotesis alternativa H_1 : $\mu > m_0$, por tanto se pueden admitir las botellas para su reutilización

Nota : El test anterior está incorporado en ref/AlternativeHypothesis

In[27]:=
```
LocationTest[pI, 175, AlternativeHypothesis → "Less", SignificanceLevel → .05]
```

Out[27]=
```
0.999282
```

B.- Estimación de μ para una población con varianza σ^2 conocida.

Sea una población con media μ, no necesariamente normal, con varianza σ^2 conocida. Tomemos una muestra con media $\bar{x}$ y tamaño n. Se trata de estimar μ para los siguientes casos:

- Intervalo de confianza bilateral

$$\bar{x} - z_{1-\alpha/2} \frac{\sigma}{\sqrt{n}} < \mu < \bar{x} + z_{1-\alpha/2} \frac{\sigma}{\sqrt{n}} \tag{7}$$

- Intervalos de confianza unilaterales

$$\mu < \bar{x} + z_{1-\alpha} \frac{\sigma}{\sqrt{n}} \tag{8}$$

$$\mu > \bar{x} - z_{1-\alpha} \frac{\sigma}{\sqrt{n}} \tag{9}$$

Ejemplo 3

Calcular el intervalo de confianza para la media de la lista del ejemplo 1. Asumir que σ =10.

Podemos resolverlo utilizando la ecuacion (7) o mas directamente utilizando NormalCI del paquete HypothesisTesting que la incorpora.

In[28]:= $\left\{\bar{x}_{pI} - \frac{10}{\sqrt{n}} z[0.975], \bar{x}_{pI} + \frac{10}{\sqrt{n}} z[0.975]\right\}$

Out[28]= {178.08, 185.92}

```
In[29]:= Needs["HypothesisTesting`"]

In[30]:= MeanCI[pI, KnownVariance → 10^2]

Out[30]= {178.08, 185.92}
```

B'.- Estimación de μ para una población con varianza σ^2 desconocida.

Sea una población con media μ, no necesariamente normal, con varianza σ^2 desconocida. Tomemos una muestra con media $\bar{x}$, desviación estandar s , tamaño n y grados de libertad $v = n$ - 1. Se trata de estimar el intervalo de confianza de μ para los siguientes casos

- Intervalo de confianza bilateral

$$\bar{x} - t_{(1-\alpha/2,\ \nu)} \frac{s}{\sqrt{n}} < \mu < \bar{x} + t_{(1-\alpha/2,\ \nu)} \frac{s}{\sqrt{n}} \tag{10}$$

- Intervalos de confianza unilaterales

$$\mu < \bar{x} + t_{(1-\alpha,\ \nu)} \frac{s}{\sqrt{n}} \tag{11}$$

$$\mu > \bar{x} - t_{(1-\alpha,\ \nu)} \frac{s}{\sqrt{n}} \tag{12}$$

Ejemplo 4

Un dosificador para el embasado de un producto farmaceutico ha estado parado 2 meses. Se vuelve a poner en marcha, se sabe que anteriormente el peso medio embasado era μ= 13.58 pero se desconoce el valor de σ . Se sospecha que el proceso puede estar desajustado. Para comprobarlo se toma una muestra de 12 unidades obteniendose para esta muestra $\bar{x}$ = 13.70 y s = 0.146. ¿Esta el proceso desajustado (apliquese un nivel de significacion del 1 %)?

Lo podemos resolver directamente aplicando la ec(10)

In[31]:= $t_{0.99,11}$ = Quantile[StudentTDistribution[12 - 1], 0.99]

Out[31]= 2.71808

In[32]:= $LS = 13.58 + t_{0.99,11} \frac{0.146}{\sqrt{12-1}}$

Out[32]= 13.6997

Es decir, la región de aceptación vendrá dado por todos los valores $\bar{x} \leq 13.6997$ y la de rechazo $\bar{x} > 13.6997$. Como 13.70 esta fuera de la región de aceptación admitiremos que el dosificador esta desajustado .

Al mismo resultado podemos llegar utizando StudentTCI del paquete HypothesisTesting (Ojo: este paquete asume un intervalo bilateral por eso el nivel de confianza es 0.01 en cada sentido, es decir, el total es 0.02)

In[33]:= ? StudentTCI

StudentTCI[μ, σ, *df*] gives a confidence interval based on Student's *t* distribution with *df* degrees of freedom. >>

In[34]:= StudentTCI[13.58, $0.146 / \sqrt{11}$, 11, ConfidenceLevel -> 0.98] [[2]]

Out[34]= 13.6997

C.- Comparación de μ_1 y μ_2, de poblaciones con σ_1^2 y σ_2^2 conocidas.

Sea las medias, μ_1 y μ_2,
de poblaciones (no necesariamente normales) con varianzas σ_1^2 y σ_2^2 conocidas.

Tomamos una muestra de cada población, con medias $\bar{x}_1$ y $\bar{x}_2$ tamaños n_1 y n_2. Calculamos

$$\sigma_d = \sqrt{\frac{\sigma_1^2}{n_1} + \frac{\sigma_2^2}{n_2}} \qquad (13)$$

Comparamos μ_1 y μ_2 para los siguientes casos:

- Hipotesis nula H_0: $\mu_1 = \mu_2$. La hipotesis se rechaza si:

$$\left| \bar{x}_1 - \bar{x}_2 \right| > z_{1-\alpha/2}\, \sigma_d \qquad (14)$$

- Hipotesis nula H_0: $\mu_1 \geq \mu_2$. La hipotesis se rechaza si:

$$\bar{x}_1 < \bar{x}_2 - z_{1-\alpha}\, \sigma_d \qquad (15)$$

- Hipotesis nula H_0: $\mu_1 \leq \mu_2$. La hipotesis se rechaza si:

$$\bar{x}_1 > \bar{x}_2 + z_{1-\alpha}\, \sigma_d \qquad (16)$$

Ejemplo 5

Una fabrica produce un alambre de cobre con un alargamiento medio $\bar{x}_1$= 27.9% y σ_1 =2.64%, obtenidos del resultado de 30 medidas. Se prueba un nuevo procedimiento de fabricación. Se toma una muestra de 13 unidades a partir de la cual obtiene una media $\bar{x}_2$ = 28.7% (se asume el mismo valor de σ). ¿Puede considerarse que el proceso sigue garantizando el mismo nivel máximo de alargamiento?.

Se trata de probar la hipotesis H_0: $\mu_1 \geq \mu_2$, para lo cual aplicamos (15). Previamente necesitamos calcular σ_d dado por (13) y continuación aplicaremos (15) aplicando un nivel de significación α=5%

In[35]:=
```
Clear["Global`*"]
```

In[36]:=
$$\sigma_d = \sqrt{\frac{\sigma_1^2}{n_1} + \frac{\sigma_2^2}{n_2}};$$

In[37]:=
```
z[c_] := Quantile[NormalDistribution[0, 1], c]
```

In[38]:= $\bar{x}_1$ = 27.9; $\bar{x}_2$ = 28.7;

In[39]:= LI = x_2 - z[0.95] σ_d /. {σ_1 → 2.64, σ_2 → 2.64, n_1 → 30, n_2 → 13}

Out[39]= 27.2581

In[40]:= x_1 <= LI

Out[40]= False

Es decir, se admite la hipótesis H_0: $\mu_1 \geq \mu_2$, y por tanto se acepta que el nuevo proceso mantiene el mismo nivel máximo de alargamiento.

C'.- Comparación de μ_1 y μ_2, de poblaciones con σ_1^2 y σ_2^2 desconocidas.

Sea las medias, μ_1 y μ_2, de poblaciones (no necesariamente normales) con varianzas σ_1^2 y σ_2^2, desconocidas pero que se pueden asumir iguales. Tomemos una muestra de cada población, con medias $\bar{x}_1$ y $\bar{x}_2$, tamaños n_1 y n_2, y calculemos:

$$s_d = \sqrt{\frac{n_1 + n_2}{n_1\, n_2} \; \frac{\sum (x_1 - \bar{x}_1)^2 + \sum (x_2 - \bar{x}_2)^2}{n_1 + n_2 - 2}} \tag{17}$$

Compararemos μ_1 y μ_2 para los siguientes casos:

- Hipotesis nula H_0: $\mu_1 = \mu_2$. La hipotesis se rechaza si:

$$|\bar{x}_2 - \bar{x}_1| > t_{(1-\alpha/2,\, \nu)}\, s_d \tag{18}$$

- Hipotesis nula H_0: $\mu_1 \geq \mu_2$. La hipotesis se rechaza si:

$$\bar{x}_1 < \bar{x}_2 - t_{(1-\alpha,\, \nu)}\, s_d \tag{19}$$

- Hipotesis nula H_0: $\mu_1 \leq \mu_2$. La hipotesis se rechaza si:

$$\bar{x}_1 > \bar{x}_2 + t_{(1-\alpha,\, \nu)}\, s_d \tag{20}$$

Ejemplo 6

Una compañia petrolera pronto debe dejar de vender gasolina que contiene tetraetil-plomo que sustituirá por otra libre de plomo. Para la gasolina sin plomo pretende conservar el mismo indice octánico. Para ello realiza 10 medidas del indice octanico en cada uno de las gasolinas. Los resultados se muestran a continuación. ¿Puede asumirse que el indice octanico de ambos tipos de gasolina es el mismo? . (Se aplicará para un intervalo de confianza del 99% y se asumira varianzas desconocidas e iguales.)

```
In[41]:= plomo = {89.5, 90.0, 91.0, 91.5, 92.5, 91.0, 89.0, 89.5, 91.0, 92.0};
         sinplomo = {89.5, 91.5 , 91.0, 89.0, 91.5, 92.0, 92.0, 90.5, 90.0, 91.0};
```

Se trata de probar la hipotesis de que las medias son iguales. Puede hacerse calculando el p-valor y aceptanto la hipotesis si p>0.05

```
In[44]:= LocationTest[{plomo, sinplomo},
          VerifyTestAssumptions → "EqualVariance", SignificanceLevel → .01]

Out[44]= 0.840909
```

Por tanto se acepta la hipotesis de que ambas gasolinas presentan el mismo indice octanico.

D.- Estimación de la diferencia de μ_1 y μ_2 de dos poblaciones con σ_1^2 y σ_2^2, conocidas.

Sean μ_1 y μ_2 las medias de dos poblaciones (no necesariamente normales) con varianzas σ_1^2 y σ_2^2, conocidas. Tomemos una muestra, con medias x_1 y x_2, tamaños n_1 y n_2, y calculamos:

$$\sigma_d = \sqrt{\frac{\sigma_1^2}{n_1} + \frac{\sigma_2^2}{n_2}} \tag{21}$$

Estimemos el intervalo de confianza para la diferencia entre μ_1 y μ_2.

- Intervalo de confianza bilateral

$$(x_1 - x_2) - z_{(1-\alpha/2)}\,\sigma_d < \mu_1 - \mu_2 < (x_1 - x_2) + z_{(1-\alpha/2)}\,\sigma_d \tag{22}$$

- Intervalos de confianza unilaterales

$$\mu_1 - \mu_2 < (x_1 - x_2) + z_{(1-\alpha)}\,\sigma_d \tag{23}$$

$$\mu_1 - \mu_2 > (x_1 - x_2) - z_{(1-\alpha)}\,\sigma_d \tag{24}$$

Ejemplo 7

Utilizando los datos del indice octánico del ejercicio 6, y asumiendo que las varianzas obtenidas son aproximadamente las de la población. ¿Puede considerarse que la diferencia entre las medias de ambos tipos de gasolina son iguales?

Podemos aplicar las ecs(21) y (22) o utilizar el paquete `ConficenceIntervals` que ya las incluye. Consideraremos un nivel de confianza del 95%.

```
In[45]:= ?? MeanDifferenceCI
```

MeanDifferenceCI[*list*$_1$, *list*$_2$] gives a confidence interval
for the difference between the population means estimated from *list*$_1$ and *list*$_2$. >>

```
Attributes[MeanDifferenceCI] = {Protected, ReadProtected}

Options[MeanDifferenceCI] =
 {ConfidenceLevel → 0.95, KnownVariance → None, EqualVariances → False}

In[46]:= MeanDifferenceCI[plomo, sinplomo, ConfidenceLevel -> 0.95 ]

Out[46]= {-1.1326, 0.932599}
```

El valor "0" (diferencia de medias) está comprendido en el intervalo anterior y por tanto se puede admitir que la deferencia de medias es nula.

D'.- Estimación de la diferencia de μ_1 y μ_2 de dos poblaciones con σ_1^2 y σ_2^2, desconocidas.

Sean μ_1 y μ_2 las medias de dos poblaciones (no necesariamente normales), con varianzas, σ_1^2 y σ_2^2, desconocidas. Tomemos una muestra de cada población, con medias $\bar{x}_1$ y $\bar{x}_2$ y tamaños n_1 y n_2, y calculamos:

$$s_d = \sqrt{\frac{n_1 + n_2}{n_1\, n_2} \frac{\sum (x_1 - \bar{x}_1)^2 + \sum (x_2 - \bar{x}_2)^2}{n_1 + n_2 - 2}} \qquad (25)$$

Estimemos el intervalo de confianza para la diferencia entre μ_1 y μ_2.

- Intervalo de confianza bilateral

$$(\bar{x}_1 - \bar{x}_2) - t_{(1-\alpha/2,\, \nu)}\, s_d < \mu_1 - \mu_2 < (\bar{x}_1 - \bar{x}_2) + t_{(1-\alpha/2,\, \nu)}\, s_d \qquad (26)$$

- Intervalos de confianza unilaterales

$$\mu_1 - \mu_2 < (\bar{x}_1 - \bar{x}_2) + t_{(1-\alpha,\, \nu)}\, s_d \qquad (27)$$

$$\mu_1 - \mu_2 > (\bar{x}_1 - \bar{x}_2) - t_{(1-\alpha,\, \nu)}\, s_d \qquad (28)$$

Ejemplo 8

Se toman muestras de dos tipos de gasolina: con plomo y otra sin plomo. Se mide el indice octánico (los valores son los del ejemplo 6). ¿Podemos considerar que la gasolina con plomo y sin plomo presentan el mismo indice octánico medio . (Se asumirá varianzas desconocidas e iguales.)

```
In[47]:= MeanDifferenceCI[plomo, sinplomo,
          ConfidenceLevel -> 0.95, EqualVariances -> True ]

Out[47]= {-1.13162, 0.931617}
```

El valor "0" (diferencia de medias) está comprendido en el intervalo anterior y por tanto se puede admitir que la deferencia de medias es nula.

F.- Estimación de la varianza

Considerese la variable aleatoria x normal. Se selecciona una muestra aleatoria x_1,..., x_n. Se calcula su varianza muestral s^2. Podemos construir el intervalo de confianza para σ^2 como sigue:

- Caso bilateral

$$\frac{(n-1)\,s^2}{\chi^2_{1-\alpha/2,\,n-1}} < \sigma^2 < \frac{(n-1)\,s^2}{\chi^2_{\alpha/2,\,n-1}} \qquad (29)$$

- Caso unilaterales

$$\sigma^2 < \frac{(n-1)\,s^2}{\chi^2_{\alpha,\,n-1}} \qquad (30)$$

$$\sigma^2 > \frac{(n-1)\,s^2}{\chi^2_{1-\alpha,\,n-1}} \qquad (31)$$

Para obtener obtener información sobre la distribucion χ^2 o ChiSquareDistribution podemos utilizar la ayuda en la forma habitual

In[48]:=
```
? ChiSquareDistribution*
```

ChiSquareDistribution[ν] represents a χ^2 distribution with ν degrees of freedom. >>

Cuya función de densidad es:

In[49]:=
```
PDF[ChiSquareDistribution[n], x]
```

Out[49]=
$$\begin{cases} \dfrac{2^{-n/2}\, e^{-x/2}\, x^{-1+\frac{n}{2}}}{\text{Gamma}\left[\frac{n}{2}\right]} & x > 0 \\ 0 & \text{True} \end{cases}$$

Ejemplo 9

Un fabricante está interesado en la determinación de la resistencia a la rotura de un determinado tipo de fibras sinteticas. Una muestra de 16 fibras da como presión de rotura la de la lista: fibra1. Por experiencias anteriores se considera que los valores están normalmente distribuidas. ¿Entre que valores estará comprendida la media y la varianza (calculadas con un 95% de intervalo de confianza)?

In[50]:=
```
fibra1 = { 48.89, 52.07, 49.29, 51.66, 52.16, 49.72, 48.00, 49.96, 49.20,
48.10, 47.90, 46.94 , 51.76, 50.75, 49.86 , 51.76};
```

El intervalo de confianza de la media esta dado por:

In[51]:=
```
MeanCI[fibra1]
```

Out[51]= {48.9838, 50.7687}

El intervalo de confianza para la varianza se obtiene aplicando la ec(19) que esta incluida en `VarianceCI`:

In[52]:=
```
VarianceCI[fibra1]
```

Out[52]= {1.53076, 6.71946}

Calcular el intervalo de confianza para la varianza del ejemplo anterior

In[53]:=
```
ChiSquareCI[0.146^2, 12 - 1, ConfidenceLevel -> 0.95]
```

Out[53]=
```
{0.0106969, 0.0614495}
```

G.- Comparación de dos varianzas.

Sea x_1~N(μ_1, σ_1^2) y x_2~N(μ_2, σ_2^2), donde μ_1, σ_1^2, μ_2, σ_2^2 son desconocidas: deseamos construir un intervalo de confianza para s_1^2 / s_2^2, donde s_1^2 y s_2^2 son varianzas muestrales obtenida, respectivamente, de n_1 y n_2 observaciones aleatorias.

- Hipotesis nula H_0: $s_1^2 = s_2^2$. Se rechaza si:

$$\frac{S_1^2}{S_2^2} < \frac{1}{F_{1-\alpha/2,\, n_2-1, n_1-1}}, \quad \text{o} \quad \frac{S_1^2}{S_2^2} > F_{1-\alpha/2,\, n_1-1, n_2-1} \qquad (32)$$

- Hipotesis nula H_0: $s_1^2 \leq s_2^2$. Se rechaza si:

$$\frac{S_1^2}{S_2^2} > F_{1-\alpha,\, n_1-1, n_2-1} \qquad (33)$$

- Hipotesis nula H_0: $s_1^2 \geq s_2^2$. Se rechaza si:

$$\frac{S_1^2}{S_2^2} < \frac{1}{F_{1-\alpha,\, n_2-1, n_1-1}} \qquad (34)$$

Ejemplo 10

Deseamos conocer si debemos asumir la hipotesis utilizada en el ejemplo 6 de varianzas iguales.

Se trata de contrastar la hipoteis H_0: $\sigma_1^2 = \sigma_2^2$frente a la la alterantiva H_1: $\sigma_1^2 \neq \sigma_2^2$Aplicamos el test para contrastar la hipotesis de la razón de varianzas $F_0 = S_1^2 / S_2^2 = 1$ frente a $F_1 = S_1^2 / S_2^2 \neq 1$. La hipotesis nula $F_0 = 1$ no puede ser rechazada para como α=0.05 ni para cualquier nivel de significacion de $\alpha \geq p = 0.73$

In[54]:=
```
VarianceEquivalenceTest[{plomo, sinplomo}]
```
Out[54]=
```
0.735885
```

H.- Estimación de la razón de dos varianzas σ_1^2 / σ_2^2

Sea x_1~$N(\mu_1, \sigma_1^2)$ y x_2~$N(\mu_2, \sigma_2^2)$. Deseamos construir un intervalo de confianza para σ_1^2 / σ_2^2. Conocemos las varianzas muestrales S_1^2 y S_2^2 obtenidas de n_1 y n_2 observaciones aleatorias. Para ello procedemos como sigue:

- Caso bilateral

$$\frac{1}{F_{1-\alpha/2,\, n_1-1, n_2-1}} \frac{S_1^2}{S_2^2} < \frac{\sigma_1^2}{\sigma_2^2} < F_{1-\alpha/2,\, n_2-1, n_1-1} \frac{S_1^2}{S_2^2} \tag{35}$$

- Casos unilaterales

$$\begin{gathered} \frac{\sigma_1^2}{\sigma_2^2} > \frac{S_1^2}{S_2^2} \frac{1}{F_{1-\alpha,\, n_1-1, n_2-1}} \text{ o} \\ \frac{\sigma_1^2}{\sigma_2^2} < \frac{S_1^2}{S_2^2} F_{1-\alpha,\, n_2-1, n_1-1} \end{gathered} \tag{36}$$

Ejemplo 11

Consideremos los valores del indice de octanaje del ejemplo anterior. Nos planteamos se podemos asumir la hipotesis utilizada de suponer varianzas iguales

In[55]:=
```
? VarianceRatioCI
```

VarianceRatioCI[*list*$_1$, *list*$_2$] gives a confidence interval
for the ratio of the population variances estimated from *list*$_1$ and from *list*$_2$. >>

In[56]:=
```
VarianceRatioCI[plomo, sinplomo]
```

Out[56]=
```
{0.31307, 5.07443}
```

Como "1" está comprendido en el intervalo anterior no rechazamos la hipótesis $S_1^2 / S_2^2 = 1$

I.- Comparación de medias en el caso de datos apareados (ISO 3301-1975(E))

Sean dos observaciones x_i e y_i de cierta propiedad o caracteristica. Se dice que son datos apareados si:

a) Corresponden a la observación del mismo elemento *i* pero bajo diferentes condiciones,

b) Corresponden a dos elementos que se pueden considerar similares en todos los aspectos para la caracteristica que es objeto del test.

Condiciones de aplicación: La serie $d_i = x_i - y_i$, son valores aleatorios independientes, con media $\overline{d}$ y media $\sigma_d \simeq s_d$, que se supone siguen aproximadamente una distribución normal.

Hipotesis nula H_0: $\overline{d} = d_0$. La hipotesis se rechaza si:

$$\left| \overline{d} - d_0 \right| > t_{(1-\alpha/2,\ \nu)} \frac{\sigma_d}{\sqrt{n}} \tag{37}$$

Hipotesis nula H_0: $\overline{d} \leq d_0$. La hipotesis se rechaza si:

$$\overline{d} < d_0 - t_{(1-\alpha,\ \nu)} \frac{\sigma_d}{\sqrt{n}} \tag{38}$$

Hipotesis nula H_0: $\overline{d} > d_0$. La hipotesis se rechaza si:

$$\overline{d} > d_0 - t_{(1-\alpha,\ \nu)} \frac{\sigma_d}{\sqrt{n}} \tag{39}$$

Samples were drawn from a pool of water at 10 randomly selected locations. Each sample was tested for zinc concentration at both the surface of the water and the bottom of the pool:

In[57]:=
```
bottom = {.430, .266, .567, .531, .707, .716, .651, .589, .469, .723};
surface = {.415, .238, .390, .410, .605, .609, .632, .523, .411, .612};
```

In[59]:=
```
LocationTest[{bottom, surface}, 0,
 {"TestDataTable", "PairedT"}, AlternativeHypothesis → "Greater"]
```

Out[59]=

	Statistic	P- Value
Paired T	4.86381	0.000445558

J.-Test de hipótesis para para paramétros binomiales

In[142]:=
```
Quit[]
```

In[1]:=
```
z[c_] := Quantile[NormalDistribution[0, 1], c]
```

A veces en vez de un valor dado por una variable continua lo que tenemos es que un item cuya inspección es se rechaza o se acepta (pasa, no pasa) sin que este se le mida un valor concreto: Por ejemplo: disponemos de un anillo a traves del cual se hace pasar el item: este pasa o no pasa. En estos casos hablamos de parámetros binomiales

Comparación de p con un valor dado p_0

Deseamos comprobar la hipótesis $H_0 : p = p_0$ (Parámetro binomial p con un valor dado p_0) frente a la hipotesis alternativa $H_1 : p \neq p_0$. Aplicaremos la aproximación a la distribución normal de la binomial. Para comprobar la hipotesis anterior tomamos una muestra aleatoria n de x items.

$$Z_0 = \frac{(x + 0.5) - n\,p_0}{\sqrt{n\,p_0\,(1 - p_0)}} \text{ si } x \leq n\,p_0 \tag{40}$$

$$Z_0 = \frac{(x - 0.5) - n\,p_0}{\sqrt{n\,p_0\,(1 - p_0)}} \text{ si } x > n\,p_0 \tag{41}$$

- La hipotesis nula $H_0 : p = p_0$ se rechaza si $|Z_0| > Z_{\alpha/2}$

Ejemplo

Un taller fabrica frenos para la industria del automovil. Deseamos comprobar si el proceso produce un rechazo del 10%. Para ello tomamos una muestra de 250 uds y encontramos que 41 son defectuosos (usar un nivel de significación α=0.05).

Se trata de comprobar $H_0 : p = 0.1$ frente a $H_1 : p \neq p_0$. Se trata de calcular Z_0 para lo que aplicamos (40)

In[2]:=
$$Z_0 = \frac{(41 - 0.5) - 250 * 0.1}{\sqrt{250 \times 0.1\,(1 - 0.1)}}$$

Out[2]= 3.26769

Calculamos $z_{1+0.05/2}$ y lo comparamos con Z_0

In[3]:=
```
Abs[Z0] > z[1 - 0.05/2]
```

Out[3]= True

Conclusión: La hipotesis nula $H_0 : p = 0.1$ deberá ser rechazada

Comparación de dos parametros binomiales: p_1 y p_2

Deseamos comprobar la hipótesis de igualdad de dos parametros binomiales $H_0 : p_1 = p_2$ frente a la hipotesis alternativa $H_1 : p_1 \neq p_2$. Aplicaremos la aproximación a la distribución normal de la binomial. Para comprobar la hipotesis anterior tomamos una muestra aleatoria de n_1 observaciones de la población 1 y de n_2 de la población 2. Supondremos que x_1 y x_2 son el número de items defectuosos en cada muestra. Los parametros binomiales para cada muestra son: $\hat{p}_1 = x_1/ n_1$ y $\hat{p}_2 = x_2/ n_2$. Entonces si la hipotesis nula es cierta: $p = p_1 = p_2$ y los dos estimadores muestrales pueden ser combinados en un estimador conjunto:

$$\hat{p} = \frac{n_1 \hat{p}_1 - n_2 \hat{p}_2}{n_1 + n_2} \tag{42}$$

El estadistico para comprobar H $_0$ es:

$$Z_0 = \frac{\hat{p}_1 - \hat{p}_2}{\sqrt{\hat{p}\,(1-\hat{p})\left(\frac{1}{n_1}+\frac{1}{n_2}\right)}} \tag{43}$$

- La hipotesis nula $H_0 : p_1 = p_2$ se rechaza si $|Z_0| > Z_{\alpha/2}$

Ejemplo

Cierto número de individuos padecen una misma enfermedad. Se dividen en dos grupos A, formado por 92 pacientes y B, por 83. Se le administra un nuevo farmaco Neopharmacol al grupo A pero no al B, por lo demás ambos grupos son tratados identicamente. El número de pacientes recuperados de la enfermedad en el grupo A son 66 y 54 en el B. ¿Mejora el Neopharmacol el tratamiento? (usar un nivel de significación α=0.05).

Hacemos la hipotesis de que el Neopharmacol no mejora el tratamiento, es decir: suponemos que la probabilidad de recuperación de ambos grupos son iguales. Se trata de comprobar la hipotesis $H_0 : p_A = p_B$, frente a la alternativa: $H_1 : p_A \succ p_B$

Para evitar ambigüedades en los símbolos que vamos a utilizar los definimos previamante utilizando el paquete `Notation`

In[4]:=
```
Needs["Notation`"]
```

In[5]:=
```
Symbolize[nA];
Symbolize[nB];
Symbolize[p̂]
```

In[16]:=
```
nA = 92
```

Out[16]= 92

In[17]:=
```
nB = 83
```

Out[17]= 83

In[18]:=
```
pA = 66 / nA // N
```
Out[18]= 0.717391

In[19]:=
```
pB = 54 / nB // N
```
Out[19]= 0.650602

In[20]:= $\hat{p} = \frac{n_A\, pA + n_B\, pB}{n_A + n_B}$ // N

Out[20]= 0.685714

- La hipotesis nula $H_0 : p_1 = p_2$ se rechaza si $|Z_0| > Z_{\alpha/2}$

  Calculamos $z_{1-0.05}$ y lo comparamos con Z_0

  Conclusión: La hipotesis nula $H_0 : p_A = p_B$ deberá ser aceptada y por tanto no podemos decir que NeoPharmacol mejore el tratamiento

K.- Intervalos de confianza para paramétros binomiales

A veces es necesario construir el intervalo de confianza 100(1-α)% para una distribución para el parametro p de una distribución binomial. Este parametro corresponde a la fracción no conforme (rechazada) de un lote o proceso. Asumimos que n es conocido. Si de una muestra aleatoria n una cantidad n_1 es no conforme entonces un estimador de la media p es $\overline{p} = \frac{n_1}{n}$. Hay varias aproximaciones para construir el intervalo de confianza. Si n es grande y p≥0.1, entonces puede utilizarse la aproximación normal, resultando un intervalo de confianza 100(1-α)%.

$$\overline{p} - Z_{\alpha/2} \frac{\sigma}{\sqrt{n}} \leq p \leq \overline{p} + Z_{\alpha/2} \frac{\sigma}{\sqrt{n}} \qquad (44)$$

con $\sigma^2 = \overline{p}\,(1 - \overline{p})$

Si n es pequeño, entonces se utiliza directamente la distribución binomial para establecer los intervalos de confianza de p. Si n es grande y p es pequeño (<1), entonces la distribución de Poisson es adecuada.

En el caso de 2 parametros binomiales, p_1 y p_2 entonces podemos construir el intervalo de confianza 100(1-α)% para las diferencias como sigue:

$$\overline{p}_1 - \overline{p}_2 - Z_{\alpha/2} \sqrt{\frac{\sigma_1}{n_1} + \frac{\sigma_2}{n_2}} \leq p_1 - p_2 \leq \overline{p}_1 - \overline{p}_2 + Z_{\alpha/2} \sqrt{\frac{\sigma_1}{n_1} + \frac{\sigma_2}{n_2}} \qquad (45)$$

con $\overline{p}_i = \frac{n_i}{n}$ y $\sigma_i{}^2 = \overline{p}_i\,(1 - \overline{p}_i)$, i= 1, 2

De una muestra aleatoria de 80 cojinetes de cigueñales 15 presentan un rugosidad mayor que la permitida en la especificación correspondiente. ¿Cuales son los límites del intervalo, calculados con un 95% de confianza, par la fracción no conforme? , ¿Puede asumirde la aproximación de la binomial a la normal?.

La fracción media de elementos no comformes es $\overline{p}$ = 15/80 ≥0.1 por tanto y n es grande, por tanto sí puede utilizarse la aproximación de la binomial por una normal para estimar el intervalo de confianza de los elementos no conformes. Por tanto puede aplicarse la ecuación (40)

In[21]:= $\overline{p} = 15/80$

Out[21]= $\frac{3}{16}$

In[22]:= intervalo = $\sqrt{\frac{\overline{p}\,(1 - \overline{p})}{n}}$ /. n → 80;

Usamos la funcion z[c] antes definida para obtener los coeficientes Z, para α = 0.05

In[23]:= {LimiteSuperior , LimiteInferior } = {$\overline{p}$ + z[0.975] intervalo , $\overline{p}$ - z[0.975] intervalo}

Out[23]= {0.273029, 0.101971}

En resumen, 0.1020≤ p ≤0.2730

Ejercicios propuestos

El resultado de los ensayos para verificar la resistencia a la rotura, en newtons, de dos muestras de cable procedentes de dos fabricantes A y B son los siguientes:

In[24]:=
```
muestraA =
  {2.297, 2.582, 1.949, 2.362, 2.040, 2.133, 1.855, 1.986, 1.642, 2.915};
muestraB = {2.286, 2.327, 2.388, 3.172, 3.158, 2.751,
   2.222, 2.367, 2.247, 2.512, 2.104, 2.707};
```

En la resolución del ejercicio aplica los criterios y el formato establecido en la ISO 2854-1976(E)

a) ¿Se puede asumir que cada muestra procede de una población normal?

Resolución

b) El fabricante A dice que su proceso produce cables con una resistencia media m_A = 2.40 newtons y una σ_A = 0.3315. El test se aplicará a una muestra de tamaño n_A =10. ¿Puede asumirse esta hipotesis?.(Utilicese un nivel de significación α = 0.05)

Resolución

In[26]:= $\bar{x}_A$ = Mean[muestraA]

Out[26]= 2.1761

In[27]:= c = 1 - α / 2 /. {α → 0.05}

Out[27]= 0.975

Calculamos, separadamente, primero el termino de la decha de la ec(1) y a continuación del de la izda, y aplicamos (1)

In[28]:= l1 = z[c] $\frac{\sigma}{\sqrt{n}}$ /. {σ → 0.3315, n → 10}

Out[28]= 0.205462

In[29]:= l2 = Abs[$\bar{x}_A$ - m_A] /. m_A → 2.40

Out[29]= 0.2239

In[30]:= l2 ≥ l1

Out[30]= True

Por tanto, se concluye que hipotesis H $_0$: $\mu = m_0$ se rechaza

Ejemplos adicionales

c) Repite el apartado anterior utilizando la desviación standard de la muestra A, s_A

d) El fabricante B indica que su proceso tiene una σ_B = 0.3112, y que para determinar la media toma una muestra de tamaño n_B =12. ¿Puede asumirse que sus cables presentan una resistencia media a la rotura similar al fabricante A.

e) Repite el apartado anterior utilizando las desviaciones standards de la muestra A, s_A, y de la muestra B, s_B.

f) Estima el intervalo central para la diferencia entre las medias de A y B, con un 95% de confianza. Asume que σ_A y σ_B son las proporcionadas por los fabricantes.

g) Repite el apartado anterior utilizando las desviaciones standards de la muestra A, s_A, y de la muestra B, s_B y asumiendo que $\sigma_A = \sigma_B$.

h) La especificación para la aceptación de los cables especifica que la varianza de la resistencia a la rotura no debe exceder $\sigma_o^2 = 0.0900$. ¿Puede asumirse, a partir de las muestras A y B, que los fabricantes cumplen con este requisito?.

i) A partir de las muestras A y B, estima el intervalo central , con un 95% de confianza, para σ_A y σ_B.

j) A partir de las muestras A y B, puede asumirse que las varianzas de ambas poblaciones son iguales $\sigma_A = \sigma_B$.

k) A partir de las muestras A y B, estima el intervalo central , con un 95% de confianza, para σ_A / σ_B.

El paquete Intervalos

En este apartado se describe el paquete Intervalos que permite calcular intervalos de confianza de la media y los de tolerancia de poblaciones, asi como el tamaño muestral.

In[31]:=
```
Needs["Estadistica`ControlCalidad`"]
```

```
ControlCalidad,  version 1.1  2015-05-19
```

In[32]:=
```
? "Estadistica`ControlCalidad`*"
```

▼ Estadistica`ControlCalidad`

g1	g2	limite-Bilat-eral	limite-Unil-ater-al	p1	ProbA-lfa	sizeS-amp-le	t1	z1

Algunas de las funciones del paquete intervalo pueden ejecutar directamente en un navegador web: http://www3.enusa.es/webMathematica/Estadistica/estadistica.htm



Intervalos de confianza de la media para varianza conocida

Sea $\bar{y}$ la media de cierta medida realizada a una muestra procedente de una población de la que se conoce su varianza que es σ^2 . Llamamos intervalo de confianza para $\bar{y}$ con nivel de confianza (1 - α) aquel que tiene una probabilidad de (1 - α)100% de contener la verdadera media μ.

Límite unilateral inferior

$$Y_p = \bar{y} - \frac{z(1-\alpha)}{\sqrt{n}}\sigma$$

Límite unilateral superior

$$Y_p = \bar{y} + \frac{z(1-\alpha)}{\sqrt{n}}\sigma$$

Límite bilateral inferior

$$Y_p = \bar{y} - \frac{z(1-\alpha/2)}{\sqrt{n}}\sigma$$

Límite bilateral superior

$$Y_p = \bar{y} + \frac{z(1-\alpha/2)}{\sqrt{n}}\sigma$$

Calcular los intervalos de confianza unilaterales y bilaterales para: $\bar{y}$ = 50, σ = 3, n = 30, y α = 0.10.

Cálculo del intervalo de confianza en muestras de po normales con varianza conocida

La siguiente función calcula el Limite Superior (LS) y el Limite inferior (LI) del intervalo de co media de una población normal de varianza conocida.

Entre n: 30 Entre alfa: 0.05

Entre la media muestral: 10.34 Entre la desviacion tipica poblacional: 1.23

Evaluate

results

El resultado muestra LS y el LI para el caso bilateral y unilateral

Nota: Pulse aquí para obtener información de ayuda

Intervalos de confianza de la media para varianza desconocida

Sea $\bar{y}$ la media de cierta medida realizada a una muestra procedente de una población de la que se no conoce su varianza, pero si la varianza muestral

Si desconocemos la varianza de la población utilizarmos la distribución *t* que la que se verifica $t = (\bar{y} - \mu)/(s/\sqrt{n})$, donde *s* es la desviación estándar muestral, con la que se obtienne los siguientes límites para los intervalos

Límite unilateral inferior

$$Y_p = \bar{y} - \frac{t(n-1,\ 1-\alpha)}{\sqrt{n}}\, s$$

Límite unilateral superior

$$Y_p = \bar{y} + \frac{t(n-1,\ 1-\alpha)}{\sqrt{n}}\, s$$

Límite bilateral inferior

$$Y_p = \bar{y} - \frac{t(n-1,\ 1-\alpha/2)}{\sqrt{n}}\, s$$

Límite bilateral superior

$$Y_p = \bar{y} + \frac{t(n-1,\ 1-\alpha/2)}{\sqrt{n}}\, s$$

Cálculo del intervalo de confianza en muestras normales con varianza desconoci

La siguiente función calcula el Limite Superior (LS) y el Limite inferior (LI) del inte media de una población normal de varianza desconocida.

Entre n: 30 Entre alfa: 0.05

Entre la media muestral: 10.34 Entre la desviacion tipica muestral: 1.2

Evaluate

results

El resultado muestra LS y el LI para el caso bilateral y unilateral

Nota: Pulse aquí para obtener información de ayuda

Cálculo de intervalos de tolerancia en poblaciones normales

Conceptos

En control de calidad a veces es necesario definir el intervalo de tolerancia (no confundir con intervalo de confianza).

Consideremos una población de la que tomamos una muestra de tamaño n en la que medimos cierta característica y (diámetro, densidad, peso, etc.). Calculamos su media $\bar{y}$ y su desviación estándar s. El cliente puede requerir que dicha característica esté dentro de un intervalo de tolerancia.

Definimos el 100p-ésimo percentil al la valor Y_p por debajo del cual está el 100p % de la población, o, lo que es equivalente, aquel valor para el que las observaciones de muestra aleatoria de la población cae con una probabilidad p. En ocasiones nos interesa estimar el valor de Y_p asociandole un intervalo de confianza, (1 - α) 100%. Ejemplo: Deseamos calcular el valor por debajo del cual está el 95% de la población con un nivel de confianza del 95%. Problemas similares consisten en calcular límites centrales dentro de los cuales está el 100p% de la población, y límites inferiores por encima del cual está 100p% de la población. Asimismo un límite superior de confianza (1 - α) 100% para el percentil 100p-ésimo es equivalente al límite superior de tolerancia por debajo del cual está al menos una proporción p de la población con una probabilidad 1 - α. Una interpretación similar cabe del límite inferior de tolerancia.

Las expresiones específicas para obtener los límites de tolerancia Y_p son las siguientes (requiere que la población sea normal)

Límite unilateral inferior

$$Y_p = \bar{y} - \frac{t'\left(n-1,\ -z(p)\sqrt{n};\ 1-\alpha\right)}{\sqrt{n}}\, s$$

Límite unilateral superior

$$Y_p = \bar{y} + \frac{t'\left(n-1,\ z(p)\sqrt{n};\ 1-\alpha\right)}{\sqrt{n}}\, s$$

Límite bilateral inferior

$$Y_p = \bar{y} - \frac{t'\left(n-1,\ -z(p)\sqrt{n};\ 1-\frac{\alpha}{2}\right)}{\sqrt{n}}\, s$$

Límite bilateral superior

$$Y_p = \bar{y} + \frac{t'\left(n-1,\ z(p)\sqrt{n};\ 1-\frac{\alpha}{2}\right)}{\sqrt{n}}\, s$$

donde t' es la distribución t – *Student* no central

`NoncentralStudentTDistribution` [ν,δ]) a noncentral Student t con ν grados de libertad y un parámetro de no centralidad δ.

In[86]:=
```
Plot[
 Evaluate@Table[PDF[NoncentralStudentTDistribution[25, δ], x], {δ, {-2, 0, 3}}],
 {x, -6, 8}, Filling → Axis]
```

Out[86]=



In[87]:=
```
Plot[
 Evaluate@Table[CDF[NoncentralStudentTDistribution[25, δ], x], {δ, {-2, 0, 3}}],
 {x, -6, 8}, Filling → Axis]
```

Out[87]=



Calculo de $k_{n,\alpha,p}$ (método exacto)

$$k_{n,\alpha,p} = \frac{t'\left(n-1,\ z(p)\sqrt{n};\ 1-\alpha\right)}{\sqrt{n}}$$

De forma exacta puede calculase aplicando:

In[88]:=
```
K1Exacto[n_, α_, p_] :=
 Module[{z}, z = Quantile[NormalDistribution[0, 1], p];
  (1 / Sqrt[n]) * Quantile[NoncentralStudentTDistribution[n - 1, z * Sqrt[n]],
    1 - α]]
```

Ejemplo: La k(95/95) unilateral para una muestra de tamaño 2 a 10 son

In[89]:=
```
Table[{r, K1Exacto[r, 0.05, 0.95 ]} , {r, 2, 10}]
```

Out[89]=
```
{{2, 26.2597}, {3, 7.6559}, {4, 5.14387}, {5, 4.20268},
 {6, 3.70768}, {7, 3.39947}, {8, 3.18729}, {9, 3.03124}, {10, 2.91096}}
```

Los valores son practicamente identicos a los encontrados en las tablas consultadas

Calculo de $k_{n,\alpha,p}$ (Método aproximado)

El método anterior requiere un consumo eleva de recursos de ordenador que lo hace poco adecuado cuando el valor *n* empieza a ser elevado (*n*>10). Una aproximación excelente a (13) puede obtenerse para el caso bilateral, que es el mas complicado aplicando la siguiente aproximación

$$k_{n,\alpha,p} \approx z[(1+p)/2]\left(\frac{(n-1)}{\chi^2(n-1;\alpha)}\right)^{1/2}\left(1+\frac{1}{2n}\right)$$

Expresado en el lenguaje del *Mathematica* es

In[90]:=
```
g2[n_, α_, p_] := Module[{z, ji2}, z = Quantile[NormalDistribution[0, 1], (1 + p)/2];
  ji2 = Quantile[ChiSquareDistribution[n - 1], α];
  z Sqrt[(n - 1)/ji2] (1 + 1/(2 n))]
```

Ejemplo: La k(95/95) bilateral para una muestra de tamaño *n* = 50 y otra para *n* = 1000 son

In[91]:=
```
g2[50, 0.05, 0.95]
```

Out[91]= 2.37889

In[92]:=
```
g2[1000, 0.05, 0.95]
```

Out[92]= 2.03608

Estos valores muestran una excelente aproximación para los encontrados en tablas

Tambien podemos aplicar una aproximación para el caso unilateral cuando el el valor *n* empieza a ser elevado (*n*>10).

$$k_{n,\alpha,p} = \frac{t'\left(n-1,\ z(p)\sqrt{n};\ 1-\alpha\right)}{\sqrt{n}} \approx \left(z(p)+\left[z^2(p)-a\,b\right]^{1/2}\right)\Big/ a$$

donde

$$a = 1 - [z(1-\alpha)]^2 \big/ [2(n-1)]$$
$$b = [z(p)]^2 - [z(1-\alpha)]^2 \big/ n$$

Expresado en el lenguaje del *Mathematica* es

In[93]:=
```
g1[n_, α_, p_] := Module[{zp, za, a, b}, zp = Quantile[NormalDistribution[0, 1], p];
  za = Quantile[NormalDistribution[0, 1], 1 - α];
  a = 1 - za^2/(2 (n - 1));
  b = zp^2 - za^2/n;
  (zp + Sqrt[zp^2 - a b])/a]
```

In[94]:=
```
g1[10, 0.05, 0.95]
```

Out[94]=
```
2.8748
```

Los aproximación aumenta con el tamaño de la muestras: Ejemplo: Calcular la k(95/95) unilateral para n={10, 100, de 10 en 10} :

In[95]:=
```
Table[{n, g1[n, 0.05, 0.95]}, {n, 10, 100, 10}]
```

Out[95]=
```
{{10, 2.8748}, {20, 2.37835}, {30, 2.20851}, {40, 2.11721}, {50, 2.05849},
 {60, 2.01681}, {70, 1.98533}, {80, 1.9605}, {90, 1.94029}, {100, 1.92344}}
```

Criterios para determinar si una muestra está dentro del intervalo de tolerancia para un P y alfa dados

Dado un intervalo de tolerancia definido por los límites superior e inferior (LS, LI) pretendemos saber si una muestra de media $\bar{x}$ y desviación estandar *s* está comprendida en el intervalo de tolerancia para un P dado

Conocidos los LS e LI de aceptacion, y los parametros muestrales $\bar{x}$, s y *n*, se trata de calcular *p* fijado α y ver si este p es mayor o igual que un *P* dado. Para ello:

a) Hacemos Y_p= LS y Y_p= LI en (10) y obtenemos

$$\mathrm{g1}[n,\ alfa,\ pS] == \frac{LS - \bar{x}}{s} \text{ y de aquí } pS\left[\mathrm{g1}[n,\ alfa],\ \frac{LS - \bar{x}}{s}\right] = p_{LS}$$

$$\mathrm{g1}[n,\ alfa,\ 1 - pI] == \frac{\bar{x} - LI}{s} \text{ y de aquí } pI\left[\mathrm{g1}[n,\ alfa],\ \frac{\bar{x} - LI}{s}\right] = p_{LI}$$

donde p_{LS} indica la proporcion de la población comprendida entre LS y 0 y p_{LI} indica la proporcion de la población comprendida entre LI y 0 (estrictamente en vez de 0 es -∞, pero asumimos que la variable x sólo toma valores positivos).

b) La población será aceptada si se verifica que p_{LS}-p_{LI} = $p \geq P$, siendo *P* la proporción de la población que se considera aceptable.

Creamos una sentencia que nos automatice el anterior proceso (para P = 0.95 y alfa = 0.05), donde mean es la media de la muestra $\bar{x}$, size su tamaño *n* , uL le LS, uL el LI y sigma la *s* de la meustra

Aceptacion/rechazo de la poblacion (normal) muestra

La siguiente función calcula si una poblacion es aceptable o rechazable dados lc inferior (LI) de aceptación. Para ello se parte de una muestra, procedente de una tamaño de media y desviación tipica dada. El calculo se hace para una alfa y P c

Tamaño muestral: 30 Alfa (nivel de confianza 1 - alfa): 0.05 P (Proporcion 0.95

Media muestral: 10.34 Desviacion tipica muestral: 1.23 Limite superior de ace inferior de aceptación: 7.4

Evaluate

results

El resultado corresponde al caso unilateral

Nota: Pulse aquí para obtener información de ayuda

Criterios para determinar si fijado alfa, el tamaño muestral y el valor de K que porcentaje de la poblacion está dentro del intervalo de tolerancia

Coeficiente de tolerancia: Cálculo de P en K1(n, alfa, P) = Ka

Dado el factor unilateral de tolerancia K1(n, alfa, P) = Ka, donde n (número de datos), alfa (el nivel de confianza) y a (valor del coeficiente) son conocidos, la función calcular el valor de P (Probabilidad).

Entre n: 30 Entre alfa: 0.05

Entre Ka: {2.0, 2.2, 2.12}

Evaluate

results

El resultado muestra P fijado n y alfa.

Nota: Pulse aqui para obtener información de ayuda

2.	2.2	2.12
0.929747	0.949279	0.942074

Dato el tamaño muestral, el nivel de confianza y la fraccion defectuosa calcular el tamaño muestral para un muestreo simple con cero rechazos


Tamaño muestral en funcion de N, p y f

Dado N , alfa (el nivel de confianza) y f (fraccion de defectuosas) son conocidos, la función calcula el valor de n (tamaño de la muestra), para un muestreo simple con cero rechazos. La entrada permite introducir para N varios valores.

Entre N. {50, 200, 500, 1000, 2000} Entre alfa. 0.05 Entre f. 0.01

Evaluate

results

El resultado muestra n fijado f, N y alfa.

Nota: Pulse aquí para obtener información de ayuda

El problema mas frecuente en muestreo es determinar el tamaño de la muestra y el número de uds. no conformes en la muestra. Vamos a describir como realizarlo en el caso de muestreos simples.

Sea una población de tamaño N, del que desconocemos el número de uds defectuosas. La probabilidad P_a de que haya r o menos defectuosas nos viene dada por la distribución hipergeométrica.

$$P_a = \sum_{x=0}^{r} \frac{\binom{D}{x}\binom{N-D}{n-x}}{\binom{N}{n}}$$

donde

D Número de piezas defectuosas en el lote (desconocidas).
f Fraccion de uds defectuosas: $f = D/N$, f es desconocida, se toma como valor la fracción máxima, p_A, de la población que admitiriamos que este formada por uds fuera de los límites de tolerancia.
α = nivel de significación (Probabilidad de cometer un error de tipo I que estamos dispuesto a asumir).
n = tamaño muestral

En el caso particular de que el criterio de aceptación sea c = 0

$$P_a = \frac{\binom{(1-f)N}{n}}{\binom{N}{n}},\ \text{entonces fijado Pa podemos calcular } n$$

Test de normalidad

EHE incluye un conjunto de test de normalidad. Es de destacar el test de D' Agostino que aplica a muestras con n grande

In[33]:=
```
Needs["Estadistica`NormalityTests`"]
```

In[34]:=
```
? "Estadistica`NormalityTests`*"
```

▼ Estadistica`NormalityTests`

DAgostinoP-earsonKS-quareTest	KurtosisM	NormalityTe-sts	Omnibus	Significance	Urzua
JarqueBera	multinormal-TestStatist-ic	normalityTe-stStatistic	Residuals	SkewnessM	

Ejemplos: Normalilty test: JarqueBera, Urzua, DAgostino

testdata1 is a vector of uniformly distributed random numbers between zero and one; testdata2 is a vector of numbers drawn from the Normal(3,1) distribution.Therefore the tests should reject normality for testdata1 and   not reject for testdata2.

To no reject the normality of the data Crit Stat ≤Crit. Value

In[104]:=
```
testdata1 = RandomReal[{0, 1}, 1000];
```

In[105]:=
```
testdata2 = RandomReal[NormalDistribution[3, 1], 1000];
```

In[106]:=
```
normalityTestStatistic[testdata1, Method -> JarqueBera, Significance -> 0.05]
```

Out[106]//TableForm=

Test Stat	N for crit	Crit.Value at 0.05
57.1959	1000.	5.99146

In[107]:=
```
normalityTestStatistic[testdata2, Method -> JarqueBera, Significance -> 0.05]
```

Out[107]//TableForm=

Test Stat	N for crit	Crit.Value at 0.05
1.27298	1000.	5.99146

In[108]:=
```
normalityTestStatistic[testdata2, Method -> Urzua, Significance -> 0.01]
```

Out[108]//TableForm=

Test Stat	N for crit	Crit.Value at 0.01
1.34301	1000.	9.21034

Using DAgostinoPearsonKSquareTest the Normally distributed data should tend to have pvalues that are not too small. At least should be requirer PValue ≥ 0.05.

In[109]:=
```
DAgostinoPearsonKSquareTest[testdata1]
```

Out[109]=
```
{KSquareStatistic → 592.314, PValue → 0.}
```

In[110]:=
```
DAgostinoPearsonKSquareTest[testdata2]
```

Out[110]=
```
{KSquareStatistic → 1.40895, PValue → 0.494368}
```

Ejemplos: Multinormality Test : Omnibus, KurtosisM, SkewnessM

We will simulate a MultinormalDistribution

In[35]:=
```
mu = {2, 3, 4};
```

In[36]:=
```
sigma = {{1, 1/2, 1/3}, {1/2, 1/3, 1/4}, {1/3, 1/4, 1/5}};
```

In[37]:=
```
multi1 = RandomReal[MultinormalDistribution[mu, sigma], 100];
```

In[38]:=
```
multi = Round[100 * multi1];
```

To no reject the normality of the data Crit Stat ≤Crit. Value

In[39]:=
```
multinormalTestStatistic[Transpose[multi],
  Method -> Automatic, Significance -> 0.9]
```

Out[39]//TableForm=

Automatic	Test Stat	N for crit	Crit.Value at 0.9
4.03947		100	11.12

In[40]:=
```
multinormalTestStatistic[Transpose[multi], Method -> Omnibus, Significance -> 0.9]
```

Out[40]//TableForm=

Omnibus	Test Stat	N for crit	Crit.Value at 0.9
4.03947		100	11.12

In[41]:=
```
multinormalTestStatistic[Transpose[multi],
  Method -> KurtosisM, Significance -> 0.9]
```

Out[41]//TableForm=

KurtosisM	Test Stat	N for crit	Crit.Value at 0.9
2.68524		100	6.07

In[42]:=
```
multinormalTestStatistic[Transpose[multi],
  Method -> SkewnessM, Significance -> 0.9]
```

Out[42]//TableForm=

SkewnessM	Test Stat	N for crit	Crit.Value at 0.9
1.35423		100	6.27

Ejemplo inclinación cabezales

In[43]:=
```
inclinacion1 = {{2.4, 0.3}, {1.36, 0.48}, {0.48, 1.94}, {0.6, 0.38}, {1.2, 0.15},
    {0.56, 1.15}, {0.5, 0.2}, {2.08, 0.62}, {0.5, 0.7}, {0.62, -0.5}, {0.88, 0.56},
    {0.6, -0.7}, {0.68, 0.18}, {1.6, 0.68}, {1.48, 0.18}, {0.68, 0.4},
    {0.58, 1.48}, {0.2, 0.52}, {0.34, 0.42}, {0.56, 0.22}, {1.74, 0.32},
    {0.16, 0.58}, {1.84, 0.22}, {0.86, 0.24}, {0.4, -0.16}, {1.12, 0.64},
    {0.26, 0.3}, {0.18, 0.22}, {0.88, 0.28}, {1.6, 0.2}, {0.5, -0.52},
    {1.5, -0.6}, {0.92, 0.16}, {1.7, -0.6}, {0.7, 0.6}, {0.84, 0.18},
    {0.32, 0.14}, {0.6, -0.4}, {0.2, -0.3}, {2.54, 0.38}, {0.4, 0.5},
    {1.5, 0.48}, {0.72, 0.14}, {1.54, 0.34}, {2.52, -0.18}, {2.04, 0.44},
    {0.54, 1.74}, {1.5, -1.2}, {0.88, 0.18}, {1.74, 0.72}, {0.4, 0.5},
    {1.58, 0.64}, {0.52, 0.6}, {0.65, 0.52}, {2.6, -1.}, {2.22, 0.44}};
```

```
In[44]:= inclinacion2 = {{0.9, 2.54}, {-0.4, 2.}, {1.5, 0.32}, {0.86, 2.04}, {1.38, 2.34},
    {1., 1.4}, {1.28, 2.54}, {0.6, 0.7}, {0.88, 2.18}, {0.75, -0.7},
    {1.44, 2.12}, {1.62, 1.54}, {0.54, 0.76}, {0.64, 0.7}, {1.94, 2.12},
    {1.9, 1.9}, {1.64, 1.1}, {1.4, 1.7}, {1.28, 2.08}, {1.1, 1.42}, {1.68, 1.2},
    {1.68, 1.2}, {3.26, -0.38}, {2.1, 2.3}, {1.3, 2.2}, {0.68, 2.54},
    {0.68, 0.74}, {1.52, 2.12}, {2.1, -3.04}, {-0.5, 2.}, {0.9, 2.8},
    {1.12, 3.02}, {1.32, 2.98}, {1.1, 1.4}, {1.62, 3.12}, {2.3, 1.2}};

In[45]:= multinormalTestStatistic[Transpose[inclinacion1],
   Method -> Automatic, Significance -> 0.9]

Out[45]//TableForm=
Automatic  Test Stat    N for crit    Crit.Value at 0.9
12.3733                 50            7.83

In[46]:= multinormalTestStatistic[Transpose[inclinacion1],
   Method -> Omnibus, Significance -> 0.9]

Out[46]//TableForm=
Omnibus  Test Stat    N for crit    Crit.Value at 0.9
12.3733               50            7.83

In[47]:= multinormalTestStatistic[Transpose[inclinacion1],
   Method -> SkewnessM, Significance -> 0.9]

Out[47]//TableForm=
SkewnessM  Test Stat    N for crit    Crit.Value at 0.9
5.32233                 50            4.64

In[48]:= multinormalTestStatistic[Transpose[inclinacion1],
   Method -> KurtosisM, Significance -> 0.9]

Out[48]//TableForm=
KurtosisM  Test Stat    N for crit    Crit.Value at 0.9
7.05093                 50            4.07

In[49]:= multinormalTestStatistic[Transpose[inclinacion2],
   Method -> Automatic, Significance -> 0.9]

Out[49]//TableForm=
Automatic  Test Stat    N for crit    Crit.Value at 0.9
66.7978                 20            8.39

In[50]:= multinormalTestStatistic[Transpose[inclinacion2],
   Method -> Omnibus, Significance -> 0.9]

Out[50]//TableForm=
Omnibus  Test Stat    N for crit    Crit.Value at 0.9
66.7978               20            8.39

In[51]:= multinormalTestStatistic[Transpose[inclinacion2],
   Method -> SkewnessM, Significance -> 0.9]

Out[51]//TableForm=
SkewnessM  Test Stat    N for crit    Crit.Value at 0.9
20.6641                 20            4.79

In[52]:= multinormalTestStatistic[Transpose[inclinacion2],
   Method -> KurtosisM, Significance -> 0.9]

Out[52]//TableForm=
KurtosisM  Test Stat    N for crit    Crit.Value at 0.9
46.1337                 20            4.22
```